The Diatomic Anharmonic Oscillator according to Matrix Mechanics

M.B. Monagan and J. F. Ogilvie

Abstract

We determine the energies of states of the diatomic anharmonic oscillator by matrix mechanics in its original form as developed by Heisenberg, Born and Jordan using perturbation theory in successive orders. We present exact formulae for the second-, fourth-, and sixth-order contributions to the energy which were computed with Maple. The calculations involve matrices of finite rank with symbolic entries. We include the Maple programs.

1 Introduction

Pioneer quantum mechanics [1] consists of matrix mechanics [2, 3, 4] and wave mechanics [5]. Both are based [6] on the following commutation law: p_m q_n − q_m p_n = −^{i h δm, n}_{2 π} Two natural algebraic objects that have this non-commutative property are differential operators and matrices. Wave mechanics is based on differential operators such that p_n is replaced by −^{i h ( ∂∂qn .)} 2 π The starting point for any wave-mechanical calculation is Schroedinger’s equation, independent of time, that has the form [5]

H ψ_k = E_k ψ_k.

in which H is a hamiltonian operator, and ψ_k is a wave function for state k. E_k is the total energy of the system in state k, which is the quantity of interest. In typical applications the hamiltonian operator consists of two contributions, potential energy V(q_n) and kinetic energy^{pm2}_{2 μ} with μ the effective mass of the system. With a known hamiltonian for a particular physical problem (e.g. for a diatomic oscillator such as HCl ) a solution of the problem requires one to find ψ_k and eigenvalues E_k with certain boundary conditions. For simple systems one can find analytic solutions, but for more complicated systems perturbation techniques and numerical methods are the norm.

Before Schroedinger [5], in 1926, inspired by de Broglie’s association of a wave with a moving particle, developed wave mechanics, Heisenberg [2] had already discovered that quantities describing physical systems obeyed the law of matrix multiplication. The first application of matrix mechanics was a linear harmonic oscillator, for which the hamiltonian matrix is diagonal with diagonal elements being the discrete energies E_k. With Jordan’s mathematical expertise, Heisenberg and Born [3, 4] developed matrix mechanics into a fully fledged physical theory. First Schroedinger [7], then others, proved that the two approaches to quantum mechanics are exactly equivalent, that is, the energies E_k computed with either method must be the same. However, because physicists were more familiar with differential equations than with matrix algebra, when Schroedinger announced his theory physicists adopted the wave-mechanical approach and interest in matrix mechanics declined.

Our objectives in this paper are to show how matrix mechanics can be applied to a significant physical problem, namely a vibrating diatomic molecule modeled as an anharmonic oscillator, and to demonstrate how symbolic computation can extend to higher orders exact calculations done previously by hand. In comparison with wave mechanics, the level of mathematics required, namely elementary linear algebra, is within the grasp of university freshmen. In contrast, wave mechanics requires maturity in partial differential equations. The anharmonic oscillator cannot be solved exactly; hence in an approximate treatment we apply perturbation theory and develop the solution as a series in λ.

2 Matrix Mechanics with Perturbation Theory

In matrix mechanics, every physical quantity is represented by a matrix [4], which are here of infinite rank. The general theory [6] to solve a problem in matrix mechanics is to transform, by a unitary transformation U, the hamiltonian matrix H into diagonal form E, as expressed in this matrix equation U H = E U. Here the matrix H is diagonal for the harmonic oscillator, in which case U is the identity matrix; thus E = H. As for the anharmonic oscillator H is no longer diagonal, the solution of the problem requires explicit diagonalization. Because we employ perturbation theory [6], we develop each matrix as a series in a parameter λ. The term of order zero captures the harmonic oscillator. Thus H = H^{(0)} + λ H^{(1)} + λ^2 H^{(2)} + ... U = U^{(0)} + λ U^{(1)} + λ^2 U^{(2)} + ... E = E^{(0)} + λ E^{(1)} + λ^2 E^{(2)} + ...

The matrix H is a known quantity; it is determined by the nature of the physical problem that is in our application a diatomic anharmonic oscillator in one dimension. The E matrix must be diagonal. The U matrix is unitary; thus U^t U = 1. Thus to first order

(U^{(0)} + λ U^{(1)}) (H^{(0)} + λ H^{(1)}) = (E^{(0)} + λ E^{(1)}) (U^{(0)} + λ U^{(1)})

Expanding this to first order yields U^{(0)} H^{(0)} + (U^{(0)} H^{(1)} + U^{(1)} H^{(0)}) λ + O(λ^2) = E^{(0)} U^{(0)} + (E^{(0)} U^{(1)} + E^{(1)} U^{(0)}) λ + O(λ^2)

Thus the solution for E^{(0)} must satisfy

U^{(0)} H^{(0)} = E^{(0)} U^{(0)}.

As H^{(0)} is diagonal, we have that E^{(0)} = H^{(0)} and U^{(0)} is the identity matrix. Replacing U^{(0)} by the identity matrix in the series we obtain this equation for the coefficient in λ

U^{(1)} H^{(0)} + H^{(1)} = E^{(0)} U^{(1)} + E^{(1)} .

Here E^{(0)}, H^{(0)}, H^{(1)} are known quantities and E^{(1)} must be diagonal. Because H^{(0)} = E^{(0)}, and H^{(0)} is diagonal, the diagonal entries of U^{(1)} H^{(0)} are the same as E^{(0)} U^{(1)}; hence the diagonal entries of E^{(1)} equal those of H^{(1)}, which happen to be zero. For off-diagonal entries of the matrix U one can also solve directly. One obtains linear equations (H^{(0)})_{j, j} (U^{(1)})_{i, j} + (H^{(1)})_{i, j} = (U^{(1)})_{i, j} (H^{(0)})_{i, i} The diagonal entries of U^{(1)} are determined from a condition U^T U = 1 as follows

(U^{(0)} + λ U^{(1)} + λ^2 U^{(2)} + ...)^T (U^{(0)} + λ U^{(1)} + λ^2 U^{(2)} + ...) = 1,

which, to second order, is

(U^{(0)})^T U^{(0)} + ((U^{(0)})^T U^{(1)} + (U^{(1)})^T U^{(0)}) λ + O(λ^2) = 1 .

As U^{(0)} is the identity matrix, the coefficient of λ simplifies to

U^{(1)} + (U^{(1)})^T = 0.

This condition requires diagonal entries to be zero. The approach taken here of solving first for diagonal entries of E and off-diagonal entries of U, then for diagonal entries of U using the unitary condition, holds equally for terms of higher order. We note explicitly in calculations that follow how this method works for the next order. In summary, given H, U, E to order K − 1, and H^{(K)}, we are able to calculate E^{(K)} and U^{(K)} . To complete the method we here define H for the diatomic anharmonic oscillator.

A general hamiltonian for a system of effective mass μ in one dimension is

H =^{p2}_{2 μ} + V(x)

For an anharmonic oscillator we express the potential energy as

V(x) = k_2 x^2 + k_3 x^3 + k_4 x^4 + ...

In this case

H^{(0)} =^{p2}_{2 μ} + k_2 x^2, H^{(1)} = k_3 x^3, H^{(2)} = k_4 x^4, ... .

The displacement coordinate x becomes a matrix X in subsequent calculations. Elements of X are defined in a solution for the harmonic oscillator that we take as a starting point for this calculation. The matrix X is symmetric with two non-zero bands given by X_{m, m+1} = A √ m + 1 for m = 0, 1, 2, ... in which A = √_h 2 μ ω0 , h is Planck’s constant, ω0 is the harmonic frequency. The matrix

H^{(0)} is equal to E^{(0)}, which is the solution for the harmonic oscillator; it is a diagonal matrix with diagonal entries H^{(0)}_{m, m} = (m +^1_2 ) ω for m = 0, 1, 2, ... in which ω = ω_0 h. This condition defines a sequence H^{(1)}, H^{(2)}, ....

3 Computations

To calculate terms of order k in a series for the matrices E and U, we proceed as follows. First we construct the matrix H^{(K)} to a given rank. Next we create a matrix equation for the coefficient of λ^K by expanding the equation U H = E U symbolically as a power series in λ . We insert all known matrices truncated to a specified rank and solve this equation for E^{(K)}and off-diagonal entries of U^{(K)} . Then, to solve for diagonal entries of U^{(K)}, we employ an equation U^T U = 1 . We expand it as a power series in λ symbolically and select the coefficient of λ^K . We insert all known quantities calculated up to this point and solve for diagonal entries of U^{(K)}. After solving for E^{(K)} and U^{(K)}, one can proceed to calculate terms of order K + 1.

3.1 Setting up Maple

The following statement tells Maple that λ is a scalar quantity.

> restart; constants := constants, lambda;

constants := false, γ, ∞, true, Catalan, FAIL, π, λ The following Maple program is used to simplify matrices of formulae; it factors out any common factor present in each entry of a matrix.

factormatrix := proc(A) local multigcd, B, L, N, D, G;
  multigcd := proc(a)
  local i,g;
    g := a[1];
    for i from 2 to nops(a) while g <> 1 do
      g := frontend(gcd,[g,a[i]]);
    od;
    g
  end;
  B := map(factor,A);
  L := {op(map(op,convert(B,listlist)))} minus {0};

  if nops(L) = 0 then RETURN( eval(B) ) fi;
  N := multigcd(map(numer,L));
  if N <> 0 then N := N / icontent(N) fi;
  if N = 0 then RETURN(eval(B)) fi;
  D := multigcd(map(denom,L));
  if D <> 0 then D := D / icontent(D) fi;
  if D = 0 then RETURN(eval(B)) fi;
  G := N/D;
  if G = 1 or G = -1 then RETURN(eval(B)) fi;
  G * map( unapply( x/G, x ), B )
end:

The following command is used to extract a submatrix from a matrix.

> with(linalg,submatrix); [submatrix ]

3.2 Constructing the matrix equations

The construction of these two matrix equations is done in the following Maple program that assigns coefficients of λ^k to variables G.k and UC .k up to a specified order k = D . This part of the computation is truly “symbolic” in that equations are in terms of unknown matrices. Later we fix the rank of these matrices, fill in known entries and solve for unknown entries.

makeEquations := proc(D) local i,H,U,E,G,Ut,UC;
  H := add(H.i*lambda^i,i=0..D);
  U := 1+add(U.i*lambda^i,i=1..D);
  E := H0 + add(E.i*lambda^i,i=1..D);
  G := expand(U &* H - E &* U);
  G := algsubs(lambda^(D+1)=0, G);
  Ut := 1 + add( transpose(U.i)*lambda^i, i=1..D );
  UC := expand( U &* Ut - 1 );
  UC := algsubs(lambda^(D+1)=0, UC);
  for i from 1 to D do
    G.i := coeff(G,lambda,i);
    UC.i := coeff(UC,lambda,i);
  od;
  RETURN();
end:
   > makeEquations(3);

For instance, here are linear coefficients as determined previously > G1; UC1;

H1 − E1 + (U1 &∗ H0 ) − (H0 &∗ U1 ) transpose(U1 ) + U1 and here are quadratic coefficients

> G2; UC2;

−(E1 &∗ U1 ) + H2 + (U1 &∗ H1 ) + (U2 &∗ H0 ) − E2 − (H0 &∗ U2 ) transpose(U2 ) + U2 + (U1 &∗ transpose(U1 )) That the matrix H^{(0)} is diagonal makes the algorithm work so simply. In the linear coefficient above, the quantity U^{(1)} H^{(0)} − H^{(0)} U^{(1)} has diagonal entries that are zero. Therefore diagonal entries of E^{(1)} are equal to those in H^{(1)}. In the quadratic coefficient, the unknown quantities are U^{(2)} H^{(0)} − H^{(0)} U^{(2)} and E^{(2)} . Diagonal entries of the first expression are again zero; hence diagonal entries of E^{(2)} are given by diagonal entries of a matrix H^{(2)} + U^{(1)} H^{(1)} − E^{(1)} U^{(1)} . The same observation applies for terms of greater order that have a general form E^{(K)} + U^{(K)} H^{(0)} − H^{(0)} U^{(K)} + ... = 0 . Diagonal entries of E^{(K)}are simply determined and off-diagonal entries of U^{(K)} yield simple linear equations. Here is the cubic coefficient

> G3; UC3;

(U1 &∗ H2 ) + (U3 &∗ H0 ) − E3 + (U2 &∗ H1 ) − (H0 &∗ U3 ) + H3 − (E1 &∗ U2 ) − (E2 &∗ U1 ) (U1 &∗ transpose(U2 )) + transpose(U3 ) + (U2 &∗ transpose(U1 )) + U3

3.3 Constructing the H matrices

The following program constructs the matrix H^{(K)} for K>0. Recall that all matrices have formally infinite rank. A parameter N specifies the rank of a truncated matrix. To obtain H^{(K)} correct to N rows and N columns we must compute X^{(K+2)} and we need to begin with N+K rows and columns for X.

makeH := proc(K,N) local m,X,H;
X := matrix(N+K,N+K,0);
for m from 0 to N+K-2 do
    X[m+1,m+2] := A*sqrt(m+1);
    X[m+2,m+1] := X[m+1,m+2];
od;
H := evalm( k.(K+2) * X^(K+2) );
submatrix(H,1..N,1..N);
end:

Here is the matrix H^{(1)} truncated to rank 6.

> H1 := makeH(1,6): factormatrix(H1);

k3 A^3

0 3 0 √ 2 √ 3 0 0

3 0 6 √ 2 0 2 √ 2 √ 3 0 0 6 √ 2 0 9 √ 3 0 2 √ 3 √ 5 √ 2 √ 3 0 9 √ 3 0 24 0 0 2 √ 2 √ 3 0 24 0 15 √ 5 0 0 2 √ 3 √ 5 0 15 √ 5 0 Formule for bands can be obtained by interpolation. We have H1 m, m+1 A3 k3 = 3 m √ m + 1 for 0 ≤ m and H1 m, m+3 A3 k3 = √ (m + 1) (m + 2) (m + 3) for 0 ≤ m.

Likewise, here is the matrix H^{(2)} truncated to rank 6.

> H2 := makeH(2,6): factormatrix(H2);

k4 A^4

3 0 6 √ 2 0 2 √ 2 √ 3 0 0 15 0 10 √ 2 √ 3 0 2 √ 2 √ 3 √ 5 6 √ 2 0 39 0 28 √ 3 0 0 10 √ 2 √ 3 0 75 0 36 √ 5 2 √ 2 √ 3 0 28 √ 3 0 123 0 0 2 √ 2 √ 3 √ 5 0 36 √ 5 0 183

Diagonal entries are given by H2 m, m A4 k4 = 6 m

2 + 6 m + 3 for 0 ≤ m

and the next band is given by H2 m, m+2 A4 k4 = 2 (2 m + 3) √ (m + 1) (m + 2) for 0 ≤ m

and the outermost band by H2 m, m+4 A4 k2 = √ (m + 1) (m + 2) (m + 3) (m + 4) for 0 ≤ m.

In general matrices H are symmetric and banded with K + 3 non-zero bands

and bands of zeroes between non-zero bands. Formulae for bands can be derived

on inspection and interpolation.

3.4 Solving for EK and UK

The following program calculates matrices E and U to orker K truncated to rank N.

AnHarmonic := proc(D,N)
  global U0,H0,E0,Order;
  local i,j,m,x,K,T,Z,U,E,H,eqns,vars,sols;
  makeEquations(D);
  U0 := array(1..N,1..N,identity);
  H0 := array(1..N,1..N,diagonal);
  for m from 0 to N-1 do H0[m+1,m+1] := (m+1/2)*omega od:
  E0 := H0;
  for K from 1 to D do
    print(‘Solving for the coefficient of‘, lambda^K);
    H.K := makeH(K,N);
    E.K := array(1..N,1..N,diagonal);
    U.K := array(1..N,1..N);
    U := evaln(U.K); E := evaln(E.K); H := evaln(H.K);
    printf(‘The matrix %s\n‘,H); print(factormatrix(H.K));
    printf(‘The matrix equation to be solved\n‘); print(G.K);
    Z := evalm(G.K);
    if K=1 then printf(‘The expanded matrix equation to be solved\n‘);
       print(Z) fi;
    eqns := normal(convert(Z,set));
    vars := {seq(seq(U[i,j],j=1..N),i=1..N), seq(E[i,i],i=1..N)} minus
            {seq(U[i,i],i=1..N)};
    sols := solve(eqns,vars); assign(sols);
    printf(‘The matrix %s\n‘, E); print(factormatrix(E));
    T := evalm( UC.K );
    printf(‘Condition for the next U matrix‘); print(UC.K = 0);
    eqns := normal(convert(T,set));
    vars := {seq(U[i,i],i=1..N)};
    sols := solve(eqns,vars); assign(sols);
    printf(‘The matrix %s\n‘,U); print(factormatrix(U));
  od;
end:

We calculate and display H^{(1)}, E^{(1)}, U^{(1)} then H^{(2)}, E^{(2)}, U^{(2)}. We also display the symbolic matrix equations. Note that E^{(1)} = 0, and all E matrices of odd order are 0.

> AnHarmonic(2,6);

Solving for the coefficient of , λ

The matrix H1 k3 A^3

0 3 0 √ 2 √ 3 0 0

3 0 6 √ 2 0 2 √ 2 √ 3 0 0 6 √ 2 0 9 √ 3 0 2 √ 3 √ 5 √ 2 √ 3 0 9 √ 3 0 24 0 0 2 √ 2 √ 3 0 24 0 15 √ 5 0 0 2 √ 3 √ 5 0 15 √ 5 0

The matrix equation to be solved

H1 − E1 + (U1 &∗ H0 ) − (H0 &∗ U1 )

The expanded matrix equation to be solved

[ −E1_{1, 1} , 3 k3 A^3 + U1_{1, 2} ω , 2 U1_{1, 3} ω , k3 A^3 √ 2 √ 3 + 3 U1_{1, 4} ω , 4 U1_{1, 5} ω , 5 U1_{1, 6} ω ] [ 3 k3 A^3 − U1_{2, 1} ω , −E1_{2, 2} , 6 k3 A^3 √ 2 + U1_{2, 3} ω , 2 U1_{2, 4} ω , 2 k3 A^3 √ 2 √ 3 + 3 U1_{2, 5} ω ,

4 U1_{2, 6} ω ] [ − 2 U1_{3, 1} ω , 6 k3 A^3 √ 2 − U1_{3, 2} ω , −E1_{3, 3} , 9 k3 A^3 √ 3 + U1_{3, 4} ω , 2 U1_{3, 5} ω ,

2 k3 A^3 √ 3 √ 5 + 3 U1_{3, 6} ω ] [ k3 A^3 √ 2 √ 3 − 3 U1_{4, 1} ω , −2 U1_{4, 2} ω , 9 k3 A^3 √ 3 − U1_{4, 3} ω , −E1_{4, 4} , 24 k3 A^3 + U1_{4, 5} ω

, 2 U1_{4, 6} ω ] [ − 4 U1_{5, 1} ω , 2 k3 A^3 √ 2 √ 3 − 3 U1_{5, 2} ω , −2 U1_{5, 3} ω , 24 k3 A^3 − U1_{5, 4} ω , −E1_{5, 5} ,

15 k3 A^3 √ 5 + U1_{5, 6} ω ] [ − 5 U1_{6, 1} ω , −4 U1_{6, 2} ω , 2 k3 A^3 √ 3 √ 5 − 3 U1_{6, 3} ω , −2 U1_{6, 4} ω , 15 k3 A^3 √ 5 − U1_{6, 5} ω

, −E1_{6, 6} ]

The matrix E1

0 0 0 0 0 0

0 0 0 0 0 0

0 0 0 0 0 0

0 0 0 0 0 0

0 0 0 0 0 0

0 0 0 0 0 0

The condition for the next U matrix is

transpose(U1 ) + U1 = 0

The matrix U1

k3 A^3

0 −3 0 − 1 3 √ 2 √ 3 0 0

3 0 −6 √ 2 0 − 2 3 √ 2 √ 3 0 0 6 √ 2 0 −9 √ 3 0 − 2 3 √ 3 √ 5 1 3 √ 2 √ 3 0 9 √ 3 0 −24 0 0 2 3 √ 2 √ 3 0 24 0 −15 √ 5 0 0 2 3 √ 3 √ 5 0 15 √ 5 0 ω Solving for the coefficient of , λ^2

The matrix H2

k4 A^4

3 0 6 √ 2 0 2 √ 2 √ 3 0 0 15 0 10 √ 2 √ 3 0 2 √ 2 √ 3 √ 5 6 √ 2 0 39 0 28 √ 3 0 0 10 √ 2 √ 3 0 75 0 36 √ 5 2 √ 2 √ 3 0 28 √ 3 0 123 0 0 2 √ 2 √ 3 √ 5 0 36 √ 5 0 183

The matrix equation to be solved

−(E1 &∗ U1 ) + H2 + (U1 &∗ H1 ) + (U2 &∗ H0 ) − E2 − (H0 &∗ U2 )

The matrix E2

A^4

3 k4 ω − 11 k3^2 A^2 , 0 , 0 , 0 , 0 , 0

0 , 15 k4 ω − 71 k3^2 A^2 , 0 , 0 , 0 , 0

0 , 0 , 39 k4 ω − 191 k3^2 A^2 , 0 , 0 , 0

0 , 0 , 0 , 75 k4 ω − 331 k3^2 A^2 , 0 , 0

0 , 0 , 0 , 0 , 123 k4 ω − 541 k3^2 A^2 , 0

0 , 0 , 0 , 0 , 0 , 183 k4 ω + 1145 k3^2 A^2 ω

The condition for the next U matrix

transpose(U2 ) + U2 + (U1 &∗ transpose(U1 )) = 0

The matrix U2

A^4 [ − 29 6 k3^2 A^2 , 0 , 3 2 (−2 k4 ω + 9 k3^2 A^2) √ 2 , 0 , 1 2 (−k4 ω + 7 k3^2 A^2) √ 2 √ 3 , 0 ] [ 0 , − 251 6 k3^2 A^2 , 0 , 1 2 (−10 k4 ω + 67 k3^2 A^2) √ 2 √ 3 , 0 , 1 2 √ 2 √ 3 √ 5 (−k4 ω + 11 k3^2 A^2) ] [ − 3 2 (−2 k4 ω + 3 k3^2 A^2) √ 2 , 0 , − 965 6 k3^2 A^2 , 0 , (−14 k4 ω + 121 k3^2 A^2) √ 3 , 0 ] [ 0 , 1 2 (10 k4 ω + 7 k3^2 A^2) √ 2 √ 3 , 0 , − 2459 6 k3^2 A^2 , 0 , 9 (−2 k4 ω + 17 k3^2 A^2) √ 5 ] [ 1 2 (k4 ω + 13 k3^2 A^2) √ 2 √ 3 , 0 , (14 k4 ω + 37 k3^2 A^2) √ 3 , 0 , − 5111 6 k3^2 A^2 , 0 ] [ 0 , 1 2 √ 2 √ 3 √ 5 (k4 ω + 17 k3^2 A^2) , 0 , 9 (2 k4 ω + 21 k3^2 A^2) √ 5 , 0 , − 3395 6 k3^2 A^2 ] / ω^2

To verify that all computations performed with the program AnHarmonic are correct, that is, that we have correctly diagonalized H to O(λ^3), we test that the equation U H = E U holds to O(λ^3).

> eqn := (1 + add(lambda^i*U.i,i=1..2)) &* add(lambda^i*H.i,i=0..2)
= add(lambda^i*E.i,i=0..2) &* (1 + add(lambda^i*U.i,i=1..2));

eqn :=

(1 + U1 λ + U2 λ^2) &∗ (H0 + H1 λ + H2 λ^2 = H0 ) + (E1 λ + E2 λ^2) &∗ (1 + U1 λ + U2 λ^2) > map(simplify@series, evalm(lhs(eqn)-rhs(eqn)), lambda, 3 ); O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^3) O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^4) O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^3) O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^4) O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^3) O(λ^4) O(λ^3) O(λ^4) O(λ^3) O(λ^4) We proceed to the second-order contribution to the energy; it is the matrix > map(expand,E2);

3 k4 A^4 − 11 k3^2 A^6 ω , 0 , 0 , 0 , 0 , 0

0 , 15 k4 A^4 − 71 k3^2 A^6 ω , 0 , 0 , 0 , 0

0 , 0 , 39 k4 A^4 − 191 k3^2 A^6 ω , 0 , 0 , 0

0 , 0 , 0 , 75 k4 A^4 − 331 k3^2 A^6 ω , 0 , 0

0 , 0 , 0 , 0 , 123 k4 A^4 − 541 k3^2 A^6 ω , 0 0 , 0 , 0 , 0 , 0 , 183 k4 A^4 + 1145 k3^2 A^6 ω Because we truncated the H matrices, not all entries in matrices E and U are correct. In fact only the first three diagonal entries in E^{(2)}are correct. We compare our results with those of Wu [6]; he employed the following definition for H.

H =^{p2}_{2 μ} +^{k1 x2}_{2!} +^{k2 x3}_{3!} +^{k3 x4}_{4!} + ....

We convert our results for E^{(2)}to conform to Wu’s notation.

> W2 := subs(k3=k2/3!, k4=k3/4!, omega=omega[0]*h, submatrix(",1..3,1..3));

W2 :=

1 8 k3 A^4 − 11 36 k2^2 A^6 ω_0 h 0 0 0 5 8 k3 A^4 − 71 36 k2^2 A^6 ω_0 h 0 0 0 13 8 k3 A^4 − 191 36 k2^2 A^6 ω_0 h These diagonal entries are consistent with equation (II-140) in [6] . The correction to the energy of the harmonic oscillator in Wu’s equation we obtain by interpolation from the three values in the above matrix. Using Maple, we obtain > interp([0,1,2],[W2[1,1],W2[2,2],W2[3,3]],n); − 1 72 A^4 (60 k2^2 A^2 − 18 k3 ω_0 h) n^2 ω_0 h − 1 72 A^4 (60 k2^2 A^2 − 18 k3 ω_0 h) n ω_0 h − 1 72 A^4 (−9 k3 ω_0 h + 22 k2^2 A^2) ω_0 h Two possible ways to write this expression are as a polynomial in A or as a polynomial in n +^1_2 . In Maple

> collect(",A,factor); − 1 36 k2^2 (30 n^2 + 30 n + 11) A^6 ω_0 h + 1 8 k3 (2 n^2 + 2 n + 1) A^4

> map(collect,series(",n=-1/2),A); (− 7 72 k2^2 A^6 ω_0 h + 1 16 k3 A^4) + (− 5 6 k2^2 A^6 ω_0 h + 1 4 k3 A^4) (n + 1 2 )^2

The corrections to higher order are obtained on running program AnHarmonic with matrices of sufficient rank. We present in the Appendix results for the correction terms E^{(4)} and E^{(6)} .

We end this section with some comments on the computations. First, regarding the contruction of H^{(K)} for K>0. When we first computed these matrices, we used H^{(1)}_{i, j} = k_3 ( ∑ i,j x_{m, i} x_{i, j} x_{j, n}) and H^{(2)} [i, j] = k_4 ( ∑ x_{m, i} x_{i, j} x_{j, k} x_{k, n}) etc. This procedure yields an algorithm to compute H^{(K)} of which takes time exponential in K. The algorithm based on computing the matrix power X^K is superior; it is rapid enough that there is little point in using formulae for the bands of H^{(K)}.

The linear system of equations that is constructed and solved is a diagonal system, but because the formulae involve several square roots of small integers, Maple runs much more slowly than if the coefficients were simple integers. Part of the reason for this is that Maple takes care to determine algebraic relationships between square roots, e.g. √ 6 − √ 2 √ 3 = 0. In our case, if one does not find √ 2, √ 3 and √ 6 objectionable in the answer, one can accelerate calculations by a factor 10 or more if one solves the equations “blindly”. We have written our own solver to do this.

4 Conclusion

We demonstrate that matrix mechanics can be used effectively to solve a significant problem in quantum mechanics with perturbation theory. We apply matrix mechanics in its purest form, not simply an expression of wave mechanics in matrix form [8]. Moreover we apply a perturbative approach whereby successive terms in potential energy expressed as a power series are treated in successive orders of the theory, not with the entire set of terms beyond a quadratic term in potential energy taken as a single perturbation to only first order as in alternative applications of perturbation theory. Because the matrix H^{(0)} is diagonal for a diatomic harmonic oscillator, the requisite matrix algebra is simple. Computations can consequently be made exactly and analytically to high order using a symbolic processor such as Maple, even though perturbation theory is an intrinsically approximate method. For reference we include here explicit formulae for terms in the energy of fourth order and sixth order in the Appendix.

References

[1] H. Primas, Chemistry, Quantum Mechanics and Reductionism second edition, Berlin, Springer-Verlag, 1983.

[2] W. Heisenberg, Z. Phys. 33, 879–893, 1925.

[3] M. Born and P. Jordan, Z. Phys. 34, 858–888, 1925. [4] M. Born, W. Heisenberg and P. Jordan, Z. Phys. 35, 557–615, 1926. [5] E. Schrodinger, Ann. Phys. Leipzig, 79, 361–376, 1926. [6] T.-Y. Wu, Quantum Mechanics, Singapore, World Scientific, 1986. [7] E. Schrodinger, Ann. Phys. Leipzig, 79, 734–756, 1926.

[8] W. Gordy and R. L. Cook, Appendix A in Microwave Molecular Spectra, third edition, Wiley, New York, 1984.

Appendix: correction terms of fourth and sixth

order_{> AnHarmonic(4,10):}

   > factormatrix(submatrix(E4,1..4,1..4)):
   > interp([0,1,2,3],[seq(E4[i,i],i=1..4)],n):
   > collect(",[A,k3,k4,k5,k6,omega],factor);

−30 (2 n + 1) (47 n^2 + 47 n + 31) k3^4 A^{12} ω^3

+ 36

(2 n + 1) (25 n^2 + 25 n + 19) k4 k3^2 A^{10} ω^2

+ (−10

(2 n + 1) (14 n^2 + 14 n + 13) k5 k3 ω − 2 (2 n + 1) (17 n^2 + 17 n + 21) k4^2 ω ) A^8

+ 5 (2 n + 1) (2 n^2 + 2 n + 3) k6 A^6 > evaln( E4[n,n] ) = map(collect, taylor(",n=-1/2), A, factor);

E4_{n, n} = (−1155 k3^4 A^{12} ω^3

+ 918 k3^2 k4 A^{10}

ω^2 − (190 k5 k3 + 67 k4^2) A^8 ω + 25 k6 A^6) (n + 1 2 )+ (−2820 k3^4 A^{12} ω^3

+ 1800 k3^2 k4 A^{10} ω^2 − 4 (70 k5 k3 + 17 k4^2) A^8 ω + 20 k6 A^6) (n + 1 2 )^3

> AnHarmonic(6,18):
> interp([0,1,2,3,4],[seq(E6[i,i],i=1..5)],n):
> evaln( E6[n,n] ) = map(collect, taylor(",n=-1/2), A, factor);

E6_{n, n} = (−

101479 4 k3^6 A^{18} ω^5 + 131817 4 k3^4 k4 A^{16} ω^4 − 1 4 k3^2 (29554 k5 k3 + 40261 k4^2) A^{14} ω^3 + 1 4 (11334 k4 k5 k3 + 6055 k6 k3^2 + 1539 k4^3) A^{12} ω^2 − 3 8 (630 k6 k4 + 770 k7 k3 + 369 k5^2) A^{10} ω + 315 8 k8 A^8) + (−418110 k3^6 A^{18} ω^5

+ 479970 k3^4 k4 A^{16}

ω^4 − 6 k3^2 (15910 k5 k3 + 20671 k4^2) A^{14} ω^3 + 6 (4890 k4 k5 k3 + 2845 k6 k3^2 + 569 k4^3) A^{12} ω^2 − 5 (354 k6 k4 + 546 k7 k3 + 217 k5^2) A^{10} ω + 245 k8 A^8)(n + 1 2 )^2 + (−463020 k3^6 A^{18} ω^5

+ 465300 k3^4 k4 A^{16}

ω^4

− 60 k3^2 (1302 k5 k3 + 1663 k4^2) A^{14} ω^3

+ 60

(322 k4 k5 k3 + 181 k6 k3^2 + 25 k4^3) A^{12} ω^2

− 30

(22 k6 k4 + 42 k7 k3 + 21 k5^2) A^{10} ω + 70 k8 A^8)(n + 1 2 )^4